
	Reason for change:
	Function f_IMS_MessageHeader_CheckTo fails in one particular scenario. Although such scenario does not occur in current TTCN, the function should be fixed in order to be future proof.


	
	

	Summary of change:
	Changed f_IMS_MessageHeader_CheckTo such that it differentiates between checking a request vs a response. 

	
	

	Consequences if not approved:
	No immediate consequences, but test case failures as soon as we check to-tag of incoming requests using above function.  

Change 1  
	Function name
	f_IMS_MessageHeader_CheckTo   

	Reason for change
	Both sides in a SIP dialog (UE and SS) store a local and a remote tag. When generating requests, the local tag is used to populate the from tag, and the remote tag is used to populate the to tag (if available). For responses, from tag and remote tag are just copied from the received request. Note that the local tag of the UE corresponds to the remote tag stored at the SS side. And the remote tag of the UE corresponds to the local tag stored at the SS side. 
Function f_IMS_MessageHeader_CheckTo is used to check certain incoming messages from the UE. Of interest are scenarios where the dialog already has both tags. When the incoming message is a response to a request sent by SS, the function works fine by comparing the stored remote tag (representing the UE side) to the to tag the UE sends in its response – the UE took this to tag just from the request the SS sent. And the SS had used the stored remote tag to populate the to tag in the request.
However, when the incoming message is an in-dialog request sent by the UE, and if we should decide to use f_IMS_MessageHeader_CheckTo in such scenario, this function would compare the stored remote tag to the to tag received. But it should consider the stored local tag instead of the stored remote tag.     
Note: currently the issue reported here does not cause test case failures because current TTCN has no call to f_IMS_MessageHeader_CheckTo that is used to check the to tag of an incoming request that actually carries a to tag. But as soon as we introduce such a check somewhere, we will run into the issue and fail test cases.

	Summary of change
	Replaced v_ToTagDialog by the pair v_RemoteTagDialog/v_LocalTagDialog (this is not strictly necessary but helps clarity).

Differentiate now between the incoming request being a request or a response. In case of a request compare the stored local tag to the received to tag. In case of a response, compare the stored remote tag to the received to tag.
This differentiation is indicated by a new boolean input argument. When there is no to tag coming in, this is irrelevant.
Note: there seems to be no corresponding function to check from headers. Do we have one resp should we have and use one? If yes, above explanation should be taken into consideration.

	TTCN module
	Common/IMS/IMS_Procedures_Common 

	MCC160 Comment
	Rejected: As stated in the reason for change already there is no use case for the proposed change. Furthermore MCC160 does not agree to the proposed implementation.


Before change

  function f_IMS_MessageHeader_CheckTo(MessageHeader p_MessageHeader,

                                       CheckToTag_Type p_CheckType) runs on IMS_PTC

  {

    var SemicolonParam_List v_ToParams := {};

    var template (omit) charstring v_ToTagDialog;
    var template (omit) charstring v_ToTagReceived;

    if (ispresent(p_MessageHeader.toField.toParams)) {

      v_ToParams := p_MessageHeader.toField.toParams;

    }

    select (p_CheckType) {

      case (noToTag) {

        if (match(v_ToParams, cr_SemicolonParam_List_WithAnyTag)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'To'");

        }

      }

      case (dialogToTag) {

        v_ToTagDialog := f_IMS_PTC_ImsInfo_DialogGetRemoteTag();       /* @sic R5s140017 MCC160 implementation: use remote tag sic@ */

        v_ToTagReceived := f_MessageHeader_GetToTag(p_MessageHeader);

        if (not ispresent(v_ToTagReceived)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Missing to-tag");

        }

        if (ispresent(v_ToTagDialog) and (valueof(v_ToTagDialog) != valueof(v_ToTagReceived))) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid to-tag");

        }
      }

    }

  }      
After change

  function f_IMS_MessageHeader_CheckTo(MessageHeader p_MessageHeader,

                                       CheckToTag_Type p_CheckType,

                                       boolean isRequest := false) runs on IMS_PTC

  {

    var SemicolonParam_List v_ToParams := {};
    //REMOVED var template (omit) charstring v_ToTagDialog;

    var template (omit) charstring v_RemoteTagDialog; 
    var template (omit) charstring v_LocalTagDialog; 
    var template (omit) charstring v_ToTagReceived;

    if (ispresent(p_MessageHeader.toField.toParams)) {

      v_ToParams := p_MessageHeader.toField.toParams;

    }

    select (p_CheckType) {

      case (noToTag) {

        if (match(v_ToParams, cr_SemicolonParam_List_WithAnyTag)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'To'");

        }

      }

      case (dialogToTag) {

        if (isRequest) { //we check to-tag in a request coming from UE

          v_LocalTagDialog := f_IMS_PTC_ImsInfo_DialogGetLocalTag();       /* @sic R5s140017 MCC160 implementation: use remote tag sic@ */

          v_ToTagReceived := f_MessageHeader_GetToTag(p_MessageHeader);

          if (not ispresent(v_ToTagReceived)) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Missing to-tag");

          }

          if (ispresent(v_LocalTagDialog) and (valueof(v_LocalTagDialog) != valueof(v_ToTagReceived))) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid to-tag");

          }

        }

        else { //we check to-tag in a response coming from UE

          v_RemoteTagDialog := f_IMS_PTC_ImsInfo_DialogGetRemoteTag();       /* @sic R5s140017 MCC160 implementation: use remote tag sic@ */

          v_ToTagReceived := f_MessageHeader_GetToTag(p_MessageHeader);

          if (not ispresent(v_ToTagReceived)) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Missing to-tag");

          }

          if (ispresent(v_RemoteTagDialog) and (valueof(v_RemoteTagDialog) != valueof(v_ToTagReceived))) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid to-tag");

          }

        }
      }

    }

  }     
